Отчет
Вычисление торических идеалов

Владимир Руцкий, 4057/2

22 декабря 2009 г.

1 Постановка задачи

Дана система уравнений вида:

$$\begin{aligned} y_1 &= x_1^{c_{11}} x_2^{c_{12}} x_3^{c_{13}} \cdots x_m^{c_{1m}} \\ y_2 &= x_1^{c_{21}} x_2^{c_{22}} x_3^{c_{23}} \cdots x_m^{c_{2m}} \\ &\vdots \\ y_n &= x_1^{c_{n1}} x_2^{c_{n2}} x_3^{c_{n3}} \cdots x_m^{c_{nm}} \end{aligned} \tag{1}$$

Или:

$$\begin{aligned} 0 &= -y_1 + x_1^{c_{11}} x_2^{c_{12}} x_3^{c_{13}} \cdots x_m^{c_{1m}} \\ 0 &= -y_2 + x_1^{c_{21}} x_2^{c_{22}} x_3^{c_{23}} \cdots x_m^{c_{2m}} \\ &\vdots \\ 0 &= -y_n + x_1^{c_{n1}} x_2^{c_{n2}} x_3^{c_{n3}} \cdots x_m^{c_{nm}} \end{aligned} \tag{2}$$

Эта система задаёт идеал a в кольце многочленов $K[y_1, \ldots, y_n, x_1, \ldots, x_m]$. Требуется найти подидеал $a' \subset a$, содержащий только переменные из $\{y_1, \ldots, y_n\}$.

2 Выбранный метод решения

Построим базис Грёбнера для a с помощью алгоритма Бухбергера с таким упорядочением мономов, чтобы в первую очередь из системы (2) были редуцированы мономы из $\{x_1, \ldots, x_m\}$. Таким упорядочением является лексикографический порядок, в котором $\forall i, j \quad x_i > y_j$.

3 Дополнительные исследования

Система (1) задаёт гомоморфизм Θ между двумя полями многочленов K_1 и K_2:

$$\Theta : K_1[y_1, \ldots, y_n] \to K_2[x_1, \ldots, x_m].$$

Подидеал a' в такой интерпретации задаёт $\ker \Theta$, т. к. $\forall v \in a' \quad \Theta(v) = 0$.

4 Реализация решения

Задача была решена в системе компьютерной алгебры Maxima[1], с использованием пакета grobner.

Число переменных было фиксировано на значениях $n = 7$, $m = 3$. Матрица степеней $c(n, m)$ была заполнена произвольными целыми степенями из диапазона $[0, 4]$, её можно найти в приложении с исходным кодом и результатами решения.

[1] `http://maxima.sourceforge.net/` — Maxima, a Computer Algebra System

(%i1)
```
/* toric_ideal.wxm
 * Calculating toric ideal.
 * Vladimir Rutsky <altsysrq@gmail.com>
 * 22.12.2009
 */

/* Loading required packages */
load(grobner)$
```

Loading maxima-grobner $Revision: 1.6 $ $Date: 2009/06/02 07:49:49 $

(%i2)
```
/* Generating powers coefficients matrix */
coefsMatrix: genmatrix(lambda([i, j], random(5)), 7, 3)$
display(coefsMatrix)$
```

$$coefsMatrix = \begin{bmatrix} 2 & 2 & 4 \\ 0 & 4 & 1 \\ 4 & 0 & 3 \\ 3 & 0 & 0 \\ 0 & 1 & 4 \\ 0 & 4 & 4 \\ 2 & 1 & 4 \end{bmatrix}$$

```
(%i4) /* Building system of equation corresponding
         to powers coefficients matrix */
      x_variables: []$
      y_variables: []$
      polynoms: []$
      for r: 1 thru length(coefsMatrix) step 1 do (
          monomialMultipliers: [],
          for c:1 thru length(coefsMatrix[1]) step 1 do (
              v: 'x[c],
              monomialMultipliers:
                  append(monomialMultipliers, [v^coefsMatrix[r][c]]),
              x_variables: append(x_variables, [v])
          ),
          v: 'y[r],
          polynoms:
              append(polynoms, [-v + apply("*", monomialMultipliers)]),
          y_variables: append(y_variables, [v])
      )$
      all_variables: append(x_variables, y_variables)$

      print("Source polynoms:")$
      for i: 1 thru length(polynoms) step 1 do
          display(polynoms[i])$
```

Source polynoms:

$polynoms_1 = x_1^2\, x_2^2\, x_3^4 - y_1$

$polynoms_2 = x_2^4\, x_3 - y_2$

$polynoms_3 = x_1^4\, x_3^3 - y_3$

$polynoms_4 = x_1^3 - y_4$

$polynoms_5 = x_2\, x_3^4 - y_5$

$polynoms_6 = x_2^4\, x_3^4 - y_6$

$polynoms_7 = x_1^2\, x_2\, x_3^4 - y_7$

```
(%i11) /* Calculating Grobner basis */
       poly_monomial_order:'lex$
       basis: poly_buchberger(polynoms, all_variables)$
```

(%i13)
```
/* Printing some polynoms from founded basis */
print("Basis has", length(basis), "polynoms.")$
display(basis[1])$
display(basis[2])$
display(basis[3])$
for i: 1 thru 4 step 1 do
    display(basis[1 + random(length(basis))])$
display(basis[length(basis) - 1])$
display(basis[length(basis)])$
```

Basis has 147 *polynoms.*

$$basis_1 = x_1^2\,x_2^2\,x_3^4 - y_1$$

$$basis_2 = x_2^4\,x_3 - y_2$$

$$basis_3 = x_1^4\,x_3^3 - y_3$$

$$basis_{24} = y_4^2\,y_5^3 - y_7^3$$

$$basis_{70} = y_3^2\,y_4\,y_6\,y_7 - y_1^2\,x_3^2\,y_4^3\,y_5$$

$$basis_{122} = x_3^2\,y_7^5 - y_1\,y_3^2\,y_5^3$$

$$basis_{59} = y_1^2\,y_2^2\,y_3\,y_5 - x_3\,y_4^2\,y_6^3\,y_7$$

$$basis_{146} = y_2\,y_3^5 - y_4^4\,y_7^4$$

$$basis_{147} = y_1\,x_2^{14} - x_1^2\,y_2^4$$

(%i20)
```
/* Displaying basis polynoms containing only 'y' variables */
for i: 1 thru length(basis) step 1 do (
    /* TODO: Use list magic */
    isFreeOf: 1,
    for j: 1 thru length(x_variables) step 1 do (
        if not freeof(x_variables[j], basis[i]) then (
            isFreeOf: 0,
            return(0)
        )
    ),
    if isFreeOf = 1 then
        display(basis[i])
)$
```

$basis_{30} = y_2\, y_3\, y_7 - y_4^2\, y_5\, y_6$

$basis_{32} = y_4^2\, y_5^3 - y_7^3$

$basis_{38} = y_1^3 - y_4^2\, y_5^2\, y_6$

$basis_{41} = y_2\, y_3\, y_5^2 - y_6\, y_7^2$

$basis_{63} = y_2^2\, y_3^2\, y_5 - y_4^2\, y_6^2\, y_7$

$basis_{83} = y_4^4\, y_6^3 - y_2^3\, y_3^3$

$basis_{111} = y_2^2\, y_7^4 - y_3^2\, y_6^3$

$basis_{112} = y_2^3\, y_5^2\, y_7^2 - y_3\, y_6^4$

$basis_{114} = y_2^2\, y_4^2\, y_5^2\, y_7^3 - y_2\, y_3^3\, y_5\, y_6^2$

$basis_{116} = y_2\, y_4^2\, y_5\, y_7^3 - y_3^3\, y_6^2$

$basis_{121} = y_2\, y_4^2\, y_7^5 - y_2\, y_3^4\, y_5\, y_6$

$basis_{124} = y_4^4\, y_5\, y_7^4 - y_2\, y_3^5\, y_5$

$basis_{128} = y_4^4\, y_5^2\, y_7^2 - y_3^4\, y_6$

$basis_{130} = y_2^4\, y_5^4 - y_6^5$

$basis_{131} = y_2^4\, y_5\, y_7^3 - y_4^2\, y_6^5$

$basis_{133} = y_3^3\, y_5^2\, y_6^2 - y_2\, y_7^6$

$basis_{141} = y_2^2\, y_3^5 - y_2\, y_4^4\, y_7^4$

$basis_{142} = y_1\, y_2\, y_3^5 - y_1\, y_4^4\, y_7^4$

$basis_{146} = y_2\, y_3^5 - y_4^4\, y_7^4$